EL AMSTRAD Y LOS NIÑOS

JACK WALKER

EDITORIAL NORAY, S.A.

EL AMSTRAD Y LOS NIÑOS

JACK WALKER

EL AMSTRAD Y LOS NIÑOS

EDITORIAL NORAY, S.A.
San Gervasio de Cassolas, 79
08022 Barcelona

Cualquier duda o aclaración sobre esta obra, será contestada por el departamento técnico de Editorial Noray, siempre y cuando se solicite por escrito al Apartado de correos n.º 6015, 08080 de Barcelona.

Título original: My Amstrad and me
Traducción de: Jaime Minguella

Primera edición, 1985
Depósito Legal: B.43078-1985
ISBN: 84-7486-057-1
Número de edición de E.N., 93
Printed in Spain - Impreso en España
Lito-Fisan, S.L. - Jaime Piquet, 7 - Barcelona

Índice

INTRODUCCION

Este libro es para niños y principiantes absolutos. No intenta ir más allá del mínimo necesario para comprender los principios de la programación. He intentado explicar las cosas concienzudamente y de manera tan simple como me ha sido posible. Espero haber conseguido evitar la presunción de que tras las primeras tres o cuatro páginas el lector se transformará milagrosamente en un experto.

La principal dificultad con los lenguajes de programación no consiste en la esencia sino en el detalle. Hay que comprender un montón de palabras, aprenderlas y colocarlas adecuadamente. Esto es interesante ya que un buen vocabulario puede producir sentencias maravillosas. Pero una serie de palabras no forma necesariamente un vocabulario. Este libro considera las palabras más importantes que (como en cualquier otro lenguaje) son las más simples. Se trata no solamente de las palabras sino de las ideas que dan forma a la programación.

Las operaciones se pueden desarrollar una tras otra, formando una secuencia.

Las operaciones se pueden repetir. Se pueden repetir un número de veces prefijado. O pueden repetirse hasta que ocurra algo que las detenga. Si nunca ocurre nada, continuarán repitiéndose.

Las operaciones pueden dirigirse en una u otra dirección, de acuerdo a una condición.

Las operaciones actuan en los objetos.

Un programa de ordenador es una serie de operaciones sobre objetos. Se facilitan los objetos a las operaciones. Las operaciones procesan los objetos para producir nuevos objetos. Estos últimos pueden ser procesados por otras operaciones del programa o pueden aparecer como resultado final del mismo.

Una de las ideas que este libro no trata es la de la recursión. No es que dicha idea sea particularmente difícil, sino que se utiliza en programas que son más complejos que cualquiera de los que aparecen en este libro. Una hora diaria con el libro, leyéndolo y operando con él, es un programa de trabajos suficiente para comprender cuan simples y potentes son las ideas de la programación.

Creo que el modo más adecuado de proceder a la lectura de este libro sería por parte de los niños y los padres conjuntamente.

PARECE FAMILIAR

Cuando veas por primera vez tu Amstrad es posible que te digas:"¡Parece una máquina de escribir!"

¿Has utilizado alguna vez una? Primero se coloca en ella una hoja de papel en blanco. Luego, pulsas las teclas y las palabras aparecen IMPRESAS en el papel. ¿Debe ser blanca la hoja de papel? No necesariamente. Puedes escribir con tinta roja sobre papel negro, con tinta negra sobre papel blanco, y así sucesivamente. ¿Qué ocurre cuando se utiliza una tinta que sea del mismo color que el papel? Si escribes tu nombre con tinta roja sobre un papel rojo, ¿serías capaz de verlo?

¿Has pensado en lo que sucede cuando pulsas la letra A en una máquina de escribir?



Al pulsar la tecla de la A, provocas un movimiento en una serie de palancas de la máquina, se levanta un brazo que presiona la cinta entintada y se imprime una A en el papel.

Si pulsas la tecla PRINT y, a continuación, A en el teclado de tu Amstrad, éste obedece tu orden de que imprima la letra A en la pantalla. Pero, por supuesto, utiliza un metodo diferente al de la máquina de escribir. El Amstrad contiene en su interior pastillas electrónicas, o chips, en vez de palancas y engranajes. Muestra cosas en la pantalla al igual que lo hace un aparato de televisión.

¿Está conectado tu Amstrad?

Observa el cuadrado al principio de la línea. Te indica que el Amstrad esta preparado para recibir las ordenes e instrucciones que tú teclees.



He terminado, así que espero la siguiente orden.

Escribe tu primer apellido escribiendo PRINT seguido de tu apellido. Ahora, el modo en que se le indica al Amstrad que ya has terminado una instrucción es pulsando la tecla ENTER. Por lo tanto, pulsa ENTER.

El Amstrad imprimirá solamente el número 0.



Esto no es lo que esperabas obtener.

Realmente, esto no es extraño si piensas que tu Amstrad es solamente una máquina. Tu aparato de televisión ofrece unas imágenes y sonidos maravillosos, pero es necesario que lo pongas en marcha. Debes cambiar los canales. Tienes que ajustar los controles de brillo, volumen, constraste, sonido.

Del mismo modo, las instrucciones debes dárselas al Amstrad en un lenguaje especial que pueda comprender. Este lenguaje se llama BASIC.

Una de las palabras de dicho lenguaje es PRINT.

Escribe:

```
PRINT"AMSTRAD"
```

Pulsa ENTER para indicarle al Amstrad que has terminado tu instrucción. ¿Qué ves en la pantalla? ¿Qué harías si quisieras poner tu primer apellido en vez de Amstrad?

El Amstrad tiene tres tipos de letras (según su tamaño).

Escribe PRINT"AMSTRAD" y después la tecla ENTER.

Ahora escribe MODE 0 (es el número 0, no la letra O) y pulsa ENTER. Escribe ahora PRINT "AMSTRAD" ¿Ves la diferencia?

Prueba ahora con MODE 2.

En MODE 0 caben 20 letras por línea con 16 colores distintos.

En MODE 1 caben 40 letras por línea con 4 colores distintos.

En MODE 2 caben 80 letras por línea con 2 colores distintos.

Prueba a escribir cosas en los 3 modos y verás la diferencia de grosor.

Prueba esto:

```
10 MODE 1
20 PRINT"AMSTRAD"
```

y después cambia la línea 10 así:

```
10 MODE 0
```

Lanza otra vez el programa. Repítelo con este cambio:

```
10 MODE 2
```

¿Ves la diferencia?

HACER LAS COSAS UNA VEZ Y HACERLAS MAS TARDE

Cuando estás en casa un fin de semana, puedes decidir súbitamente salir a jugar al fútbol. Despues, puedes desear leer un libro. Puedes continuar decidiendo hacer cosas a medida que vayas pensando en ellas, sin tener un plan definido. Pero tambien puedes, en vez de eso, hacer una lista de todas las cosas que quieres hacer, una tras otra. Si lo haces así, podrás mirar la lista más tarde y ver lo que tienes que hacer.

Supongamos que tu lista es ésta:

```
10 JUGAR AL FUTBOL
20 LEER UN LIBRO
30 VER LA TELEVISION
40 COMER
50 JUGAR CON EL
   AMSTRAD
60 FIN
```



¿Por qué crees que los números van de diez en diez? Vamos a pensarlo.

Supongamos que quieres darte una ducha despues de jugar al fútbol. Entonces podrías escribir:

15 TOMAR UNA DUCHA

Como puedes ver, la disposición de los números, de diez en diez, nos facilita la colocación de alguna cosa más en la lista de cosas a hacer.

Observa cómo te indicas a ti mismo que termines de hacer las cosas de la lista. Para ello colocas la instrucción de la línea 60.

Ahora, tú le puedes ordenar al Amstrad que haga las cosas de una en una:

```
PRINT"AMSTRAD"
```

No olvides pulsar la tecla ENTER tras el comando.

Pero, en vez de eso, puedes darle una lista de órdenes para que el Amstrad las obedezca:

Primero limpia la pantalla del ordenador utilizando el comando CLS. Luego escribe la línea a imprimir y finalmente el comando END.

```
10 CLS
20 PRINT"AMSTRAD"
30 END
```

No olvides pulsar la tecla ENTER tras escribir la línea 10 y, tras escribir las líneas 20 y 30.

Cuando tú le das al ordenador una lista de instrucciones a seguir, él las recuerda. Pero no las lleva a cabo inmediatamente. Seguirá las instrucciones únicamente cuando tú se lo ordenes.

Esta orden se la das con la palabra RUN.

La lista de instrucciones se llama un PROGRAMA.

Si quieres decirle al ordenador que vas a darle un nuevo programa, pulsa NEW.

Por lo tanto, pulsa NEW.
Ahora escribe este programa.

```
10 PEN 2
20 PRINT"AMSTRAD"
30 END
```

Observa que la línea 10 es diferente a la línea 10 tecleada antes.
Ahora utiliza la tecla RUN.

NUMEROS Y CARACTERES

Veamos de nuevo el programa.

```
10 PEN 2
20 PRINT"AMSTRAD"
30 END
```

Mira otra vez la línea 20:

```
20 PRINT"AMSTRAD"
```

¿Por qué no decimos simplemente AMSTRAD sin las "comillas"? Inténtalo si quieres y verás que el Amstrad no te entiende.

Esto ocurre porque el Amstrad necesita saber si estás hablándole de un carácter o de un número.

Los números se pueden sumar, o restar, o multiplicar, o dividir.

```
1 + 2 = 3
4 - 2 = 2
4 * 2 = 8 (* para el Amstrad significa multiplicar.
           Búscalo en el teclado.)
4 / 2 = 2 (/ para el Amstrad significa dividir.
           Búscalo en el teclado.)
```

Los caracteres son las cosas que no se tratan como si fueran números.

"AMSTRAD" es una serie de caracteres. No puedes tratarlo del mismo modo que tratarías un número. No puedes hacer cosas como sumarle o restarle algo.

Las "comillas" le dicen al Amstrad que se trata de caracteres.

Pero veamos 1984.

1984 parece un número pero también se puede tratar como si fuera una serie de caracteres. Depende de lo que quieras significar cuando dices 1984.

Si quieres referirte al año 1984, es una serie de caracteres. Si te refieres al dinero que tienes en el banco, entonces es un número.

Tu nombre es una serie de caracteres. No puedes multiplicar tu primer apellido por tu nombre para obtener un nuevo nombre. Pero si puedes multiplicar diez por dos para obtener veinte.

¿QUE PASA EN EL INTERIOR?

Esto, una vez más, es la lista de instrucciones de todo lo que puedes hacer en un sábado:

```
10 JUGAR AL FUTBOL
15 DARTE UNA DUCHA
20 LEER UN LIBRO
30 VER LA TELEVISION
40 COMER
50 JUGAR CON EL AMSTRAD
60 FIN
```

Escribes estas instrucciones en un pedazo de papel. ¡No puedes escribirlas en el aire! Necesitas algo –un pedazo de papel– donde escribir las instrucciones. Incluso, si escribes más instrucciones, necesitarás más papel.

Del mismo modo, cuando introduces un programa en el Amstrad, se conserva en el espacio que el utiliza para almacenar las instrucciones que tú le das en el programa. Cuanto más largo sea tu programa, más espacio necesitará.

Este espacio se encuentra en la memoria del ordenador. Puedes pensar en la memoria como si estuviera formada por pequeñas cajitas.

¿Significa esto que si mirases en el interior de estas cajitas podrías ver el mismo tipo de letras y números que escribes en un pedazo de papel?

No, nada de eso.

Toma dos pedacitos de papel en blanco. En el primero escribe un 1 en un lado y un 0 en el otro lado. Haz exactamente lo mismo con el otro pedazo de papel.

Ahora coloca los dos pedazos uno junto a otro. Si lo haces así y le das la vuelta a uno de los pedazos y, luego, al otro, obtendrás las combinaciones siguientes:

11
10
01
00

Si te imaginas que tú eres el Amstrad y que cada una de las combinaciones significa algo para tí, tendrás una idea de cómo almacena el Amstrad la información en su memoria.

Por ejemplo, puedes decir que la combinación 11 significa la letra P.

Cada vez que pulsas una tecla, el Amstrad cambia las letras en combinaciones como las anteriores. Por supuesto, antes teníamos únicamente cuatro combinaciones. Esto es porque utilizábamos solo dos pedazos de papel. Pero si coges 8 pedazos de papel en blanco y haces exactamente lo que hemos hecho antes, obtendrás 256 combinaciones diferentes. Es más que suficiente para que el Amstrad cambie lo que tú escribes en combinaciones de 0 y 1 que pueda entender.

Desde luego, tú no puedes entender el lenguaje que él utiliza en su interior. Por eso, el Amstrad, muy amablemente, lo cambia al lenguaje que tú entiendes cuando te muestra algo en la pantalla.

ALGUNAS COSAS PERMANECEN IGUALES Y OTRAS COSAS CAMBIAN

Tú naciste con una nariz. Tendrás una nariz toda tu vida (Espero). Cuando algo no cambia, se llama CONSTANTE. ¿Podrías indicar otras cosas a las que puedas llamar constantes?

Mira ahora mismo lo que hay en alguno de tus bolsillos. Quizá esté vacío. O quizá haya algo en él, puede que dinero, o un número de teléfono, o el nombre de tu estrella musical favorita. El contenido de tu bolsillo puede cambiar. Las cosas que pueden cambiar se llaman VARIABLES.

Si tu bolsillo contiene cosas que se pueden sumar, restar, multiplicar o dividir, contiene variables NUMERICAS. Por ejemplo, si en tu bolsillo hay cinco pesetas, puedes añadirles dos pesetas para obtener siete. O puedes poner en el bolsillo 15 pesetas, es decir tres veces cinco.

Si tu bolsillo contiene cosas modificables como números de teléfono, nombres o direcciones, contiene lo que el Amstrad llama variables de CADENA. Recuerda que no las puedes tratar como si fueran números. No puedes restar tu dirección de la dirección de cualquier otro para obtener una nueva dirección.

Puedes imaginar que el Amstrad tiene muchos, muchos bolsillos en la memoria. En estos bolsillos de la memoria conserva las variables numéricas o las variables de cadena. Por supuesto, tambien puede contener, en esos bolsillos de memoria, cosas que no cambian.



UN PROGRAMA MAYOR

Ahora, escribamos el mismo programa que utilizamos antes. Primero, escribe NEW y luego:

```
10 PEN 222
```

Me he equivocado, pero me he dado cuenta antes de pulsar ENTER. Tengo que corregirlo, claro, y para ello, lo único que he tenido que hacer es pulsar la tecla DEL para borrar los dos últimos números.

A continuación, escribo:

```
20 PRINT A"AMSTRAD"
```

¡Oh, no! Yo no quería que estuviera ahí esa A. Si la dejo, el Amstrad pensará que he escrito tonterías. Por lo tanto, tengo que corregirlo. Pulso la flecha a la izquierda hasta ponerme encima de la A. Después pulso CLR. A continuación pulso ENTER.

Ahora utilizo CLS y LIST. Esto ha sido pesado, pero todos cometemos errores y es por eso por lo que es útil ser capaz de corregir las líneas del programa en vez de volverlas a escribir de nuevo completamente. Esto sería todavía más pesado.

```
10 PEN 2
20 PRINT"AMSTRAD"
50 END
```

Ahora vamos a cambiar un poco este pequeño programa y lo vamos a hacer más grande.

Primero, queremos eliminar la línea 20. Lo único que hay que hacer es escribir:

```
20
```

Ahora pulsa ENTER. Este es el modo de eliminar una línea completa de programa.

Pulsa la tecla CLS y, luego, utiliza LIST.

Queremos decirle al Amstrad que reserve uno de sus bolsillos de memoria para una variable NUMERICA. Ahora todos los bolsillos son iguales. De modo que ¿cómo podemos decirle si un bolsillo debe contener un número o una variable de cadena?

Supongamos que quieres reservar uno de tus bolsillos únicamente para dinero. Un modo de hacerlo (aunque no muy bueno) consistiría en coger una etiqueta adhesiva, escribir en ella "CAJA", y pegarla en el bolsillo.



Si quieres reservar un bolsillo para poner únicamente nombres, podrías coger una etiqueta adhesiva, escribir en ella NOMBRE$ y pegarla en dicho bolsillo. El signo $ te indica que no estás tratando con números sino con caracteres. Cuando almacenas nombres estás tratando con variables de CADENA. No se puede utilizar la aritmética con los nombres.

¿Qué harías si quisieras reservar un bolsillo para guardar números de teléfono? ¿Un número de teléfono es un número que la gente puede sumar con otro número de teléfono? ¿La etiqueta adhesiva correcta para dicho bolsillo sería TELEFONO$?



VARIABLES NUMERICAS

Supongamos que tu bolsillo de CAJA empieza sin nada de dinero. Tu CAJA es 0.

Podemos decir:

CAJA = 0

Ahora, supongamos que añades 5 pesetas a lo que hay en el bolsillo de CAJA.

Podríamos decir:

```
CAJA = CAJA + 5
```

¿Cuánto hay ahora en tu bolsillo? Si CAJA empezó siendo 2 y le añadiste 7, ¿cuál es el valor de CAJA que habrá ahora en el bolsillo?

¿Qué harías si quisieras reservar un bolsillo para CANICAS? Escribe una nueva línea 20:

```
20 CAJA = 0
```

Ahora le has ordenado a tu Amstrad que reserve uno de sus bolsillos de memoria para una variable NUMERICA llamada CAJA. En el interior de este bolsillo de memoria, él pone el valor 0.

¿Cómo podemos saber lo que el Amstrad tiene en el bolsillo de memoria en donde se encuentra la variable CAJA? Al fin y al cabo, como se trata de una variable numérica, podemos modificarla sumándole, restándole, multiplicándola o dividiéndola. No queremos preocuparnos de recordar qué es lo que hay en el bolsillo de la variable, especialmente si lo que hay en él está cambiando continuamente. Preferiríamos que el Amstrad hiciese amablemente el trabajo por nosotros y nos dijera lo que queremos saber.

Es fácil. Lo único que tenemos que hacer es ordenarle al Amstrad que imprima (PRINT) en la pantalla lo que hay en CAJA.

Para ello, pongamos en la línea 40:

```
40 PRINT CAJA
```

Lanza el programa. No te sorprendas de que

aparezca un cero en la pantalla. Despues de todo, así es como ha empezado CAJA.

Supongamos que un amigo tuyo acaba de llegar y ha visto el 0 y te pregunta qué es lo que significa. Tú podrías decir: "Es la CAJA". ¿Pero por qué malgastar tu aliento si el Amstrad puede dar el mensaje en tu lugar?

Para ello, escribe en la línea 30:

```
30 PRINT"CAJA = "
```

Recuerda que si la pantalla se llena de cosas, puedes limpiarla pulsando la orden CLS. Lista el programa.

¿Ves la diferencia entre la Línea 30 y la Línea 40? Las comillas " de la Línea 30 le dicen al "AMSTRAD" que se trata de caracteres en vez de números.

Ahora nuestro programa debería ser algo así:

```
10 PEN 1
20 CAJA = 0
30 PRINT"CAJA = "
40 PRINT CAJA
50 END
```

Ahora lánzalo. Puedes ver como el Amstrad imprime el mensaje en la pantalla.

Ahora, digamos que queremos incrementar el contenido del bolsillo de memoria CAJA en un valor de 5.

Escribamos en una línea nueva, la Línea 50:

```
50 LET CAJA = CAJA + 5
```



Digámosle al Amstrad que imprima un mensaje tras esto. Escribe la línea 60:

```
60 PRINT"CAJA = "
```

Ordenémosle también al Amstrad que nos diga cómo ha cambiado la variable CAJA tras añadirle 5. Escribe:

```
70 PRINT CAJA
```

Escribe la línea 80:

```
80 END
```

Limpia la pantalla y lista el programa:

```
10 PEN 1
20 CAJA = 0
30 PRINT"CAJA = "
40 PRINT CAJA
50 CAJA = CAJA + 5
60 PRINT"CAJA = "
70 PRINT CAJA
80 END
```

¿Qué pone el Amstrad en el bolsillo de la variable numérica CAJA cuando llega a la línea 20?

¿Qué pone el Amstrad en el bolsillo de la variable numérica CAJA cuando llega a la línea 50?

Ahora lanza el programa. No olvides que cada vez que quieras ver las líneas del programa, una vez ejecutado, puedes utilizar otra vez LIST.

Volvamos a tu propio bolsillo. Supongamos que empiezas, una vez más, con el bolsillo vacío. Ahora, supongamos que le añades dos pesetas, tres veces.

CAJA = 0 Al principio no hay nada

CAJA = CAJA + 2 La PRIMERA VEZ hay 2 pesetas

CAJA = CAJA + 2 La SEGUNDA VEZ hay 4 pesetas

CAJA = CAJA + 2 La TERCERA VEZ hay 6 pesetas

Ahora, supongamos que quieres ser capaz de decirte a tí mismo que sumes a lo que hay en tu bolsillo, sin necesidad de recordar cuántas veces quieres hacerlo.

Supongamos que coges un pedazo de papel y escribes en él estas instrucciones:

```
FOR K = 1 TO 3
CAJA = CAJA + 2
NEXT K
```

Ahora supongamos que coges una etiqueta adhesiva, escribes en ella K y la pegas en otro bolsillo. Ahora, empieza a leer las instrucciones que escribiste.

La instrucción FOR K = 1 TO 3 significa que, la PRIMERA VEZ, habrá un 1 en el bolsillo K. Luego añades 2 pesetas al bolsillo CAJA. Al encontrarte con la instrucción NEXT K, sabes que el bolsillo K contiene un 2, la SEGUNDA VEZ que tienes que añadir 2 pesetas al bolsillo CAJA. Llegamos de nuevo a NEXT K, y el bolsillo K contiene 3, la TERCERA VEZ que tienes que añadir 2 pesetas al bolsillo CAJA.

Tan pronto como K llega a ser superior a 3, tú paras de añadir 2 pesetas al bolsillo CAJA. Esto es así porque la instrucción FOR K = 1 TO 3 te dice que lleves a cabo la operación solamente tres veces.

Intentémoslo de nuevo con números diferentes.

Supongamos que quieres añadir 3 pesetas cada vez a tu bolsillo CAJA. Supongamos que quieres hacerlo 4 veces.

Estas son las instrucciones que deberías escribirte para ello:

```
CAJA = 0
FOR K = 1 TO 4
CAJA = CAJA + 3
NEXT K
```

¿Cuántas veces añades 3 pesetas al bolsillo CAJA? ¿Cuántas pesetas habrá en el bolsillo CAJA tras haberle añadido 3 pesetas durante 4 veces?

Recuerda nuestro programa anterior:

```
10 PEN 1
20 CAJA = 0
30 PRINT"CAJA = "
40 PRINT CAJA
50 CAJA = CAJA + 5
60 PRINT"CAJA = "
70 PRINT CAJA
80 END
```

En la línea 50, el Amstrad añade 5 al bolsillo de la variable numérica CAJA. Ordenémosle que lo haga 20 veces.

Escribe la línea 45:

```
45 FOR K = 1 TO 20
```

Ahora escribe la línea 55:

```
55 NEXT K
```

Lista el programa. Utiliza CLS, si quieres, antes de pulsar LIST. Ahora tenemos:

```
10 PEN 1
20 CAJA = 0
30 PRINT"CAJA = "
40 PRINT CAJA
45 FOR K = 1 TO 20
50 CAJA = CAJA + 5
55 NEXT K
60 PRINT"CAJA = "
70 PRINT CAJA
80 END
```

El Amstrad hará todo lo que se encuentra entre las líneas 45 y 55 tantas veces como le indique la línea 45. Esta línea le dice que K debe ir subiendo hasta llegar a 20. Puedes cambiarlo para que K llegue hasta 10 o 30 o cualquier número que quieras. Pruébalo poniendo que K llegue hasta 1000.

Supongamos que quieres que el Amstrad le sume 7 al bolsillo de la variable numérica CAJA 40 veces. ¿Qué debería cambiar en la línea 45? Pruébalo y recuerda que en la línea 50 le has de decir que le sume 7 al bolsillo CAJA.

SALVALO

¿Recuerdas la lista de cosas que podíamos hacer en un fin de semana? Pues mira, hemos decidido romperla.



Podríamos tirarla o rasgarla y, de ese modo, destruirla. Si pulsas la tecla ESC, (2 veces) (¡pero no ahora!) en el teclado del Amstrad, cancelarás el programa. Si utilizas LIST verás de nuevo el programa en la pantalla. Supongamos que queremos utilizar la lista de nuevo algún otro fin de semana. Nos gustaría conservarla y almacenarla en algún lugar seguro. Sería una buena paliza si tuviésemos que escribir las instrucciones de la lista una y otra vez. Por eso salvamos la lista.

Nuestro Programa de Dinero para el Amstrad es algo que podríamos salvar antes de desconectar el ordenador. Recuerda, aunque algunas veces tratemos a esta hermosa maquinita como si fuese humana, lo hacemos únicamente porque tenemos imaginación. El Amstrad es solamente una máquina y, tan pronto como la desconectamos, se va a dormir y olvida todo lo que le hemos dicho. Por eso debemos salvar nuestro Programa de Dinero antes de mandar al Amstrad a dormir.

¡Pero todavía no! Solo una vez más, mira el Programa de Dinero de nuevo. Si todavía no está en el Amstrad, escríbelo. Cambia la línea 45 por:

```
45 FOR K = 1 TO 1000
```

Ahora lístalo y, luego, lánzalo. Ahora lístalo y cambia el 1000 de la línea 45 otra vez a 20.

Pero esta vez, tras escribir RUN, pulsa rápidamente la tecla ESC, 2 veces seguidas. El Amstrad abandona la ejecución de las instrucciones del Programa de Dinero.



Observa que el Amstrad está realmente intentando servirte. Te indica en qué línea del programa se ha detenido. No está mal para algo que no es más que una colección de chips.

Antes de irnos, recordemos que nuestro Amstrad no puede pensar por sí mismo. Te dice donde se ha parado porque tiene un programa en su interior que ya venía preparado cuando compraste el ordenador. Este programa preparado hace tambien cosas como el permitirte copiar tus programas en cinta.

La tecla ESC te permite detener un programa en ejecución. Pero al programa mismo no le ocurre nada. Puedes lanzarlo de nuevo o listarlo o, incluso, modificarlo. Modificar un programa se dice EDITAR un programa. Ya hemos hecho algunas ediciones. ESC te da la oportunidad de cambiar de opinión cuando estás en plena ejecución de un programa.

¡Un minuto! A veces, la gente que escribe programas no quiere que tú puedas utilizar la tecla ESC del modo que acabamos de indicar. Por eso te encontrarás que han impedido que estas teclas actuen como lo hacen normalmente. Han desactivado la tecla. Perdona por haberte hecho esperar tanto para decirte como salvar tu Programa de Dinero. El modo es éste:



Pon una cinta –una de las de 15 minutos– en el grabador. Asegúrate de que la cinta está al principio y, luego, hazla avanzar un poco de manera que realmente haya cinta magnética en el punto de grabación de tu programa y no un pedazo de plástico inservible.

Ahora escribe:

```
SAVE"DINERO"
```

Recuerda, nuestro Amstrad no sabe que has terminado de escribir una orden hasta que no pulsas la tecla ENTER. Por lo tanto, pulsa ENTER.

Mira el mensaje que el Amstrad te da en la pantalla:

```
PRESS REC and PLAY then any key
```

Esto significa: "Pulsa Record y Play en el cassette y después una tecla". Haz esto exactamente.

Algunas veces, es posible que tu programa no se grabe correctamente. Cuando esto ocurra, pídele ayuda a alguna persona mayor.

Veamos otra vez la instrucción SAVE"DINERO".

DINERO es el nombre que le hemos dado a nuestro programa. Pero podríamos darle cualquier otro nombre que quisiéramos, mientras no sea superior a 16 caracteres de longitud. En vez de DINERO, podríamos haber escrito SAVE"ALMAS". Pero es mejor darles a los programas un nombre que te recuerde para qué sirven.

Ahora que hemos aprendido cómo salvar un programa, querremos saber cómo volverlo a introducir en el Amstrad desde la cinta en que se encuentra. Por supuesto, debes tener una idea del lugar en la cinta en donde grabaste el programa. Con el cuentavueltas puedes apuntarte en qué número empieza y termina el programa.

Supongamos que un programa empieza en el 150 de tu cuentavueltas. Puedes hacer avanzar o retroceder la cinta hasta llegar al 150. A continuación, puedes empezar a cargar tu programa.

1. Acércate todo lo que puedas al principio de tu programa. Para ello, es posible que tengas que hacer avanzar la cinta.
2. Escribe LOAD y después:

"DINERO"

3. A continuación, pulsa ENTER.
4. Aprieta el botón PLAY.
5. Vuelve a pulsar ENTER.

En la pantalla verás:

LOADING DINERO BLOCK 1

Luego verás:

READY

Esto querrá decir que el programa está de nuevo en la memoria del Amstrad.

Si quieres, puedes listar el programa y modificarlo. En otras palabras, EDITARLO. O, por el contrario, puedes decirle al Amstrad que siga las instrucciones del programa, utilizando RUN.

Para terminar. Hemos salvado el programa DINERO. Podemos cargarlo más tarde en cualquier momento que queramos. Hagamos una pequeña pausa.

VISUALIZACION

Espero que hayas salvado en cinta el Programa de Dinero. Si lo has hecho, cárgalo y lístalo. Sea como sea, helo aquí de nuevo:

```
10 PEN 1
20 CAJA = 0
30 PRINT"CAJA = "
40 PRINT CAJA
45 FOR K = 1 TO 20
50 CAJA = CAJA + 5
55 NEXT K
60 PRINT"CAJA = "
70 PRINT CAJA
80 END
```



Lanza el programa.

Veamos cuidadosamente lo que el Amstrad representa, o visualiza, en la pantalla. Observa que el mensaje CAJA aparece en una línea de la pantalla y el número en la línea siguiente.

Quizá prefiramos tener el número en la misma línea de pantalla en que aparece CAJA. ¿Cómo podemos hacerlo?

Por supuesto, tendremos que modificar o editar el programa, ya que ahora queremos que haga algo diferente.

Utiliza la tecla SHIFT y la flecha hacia arriba hasta llegar al número 30.

Pulsa la tecla COPY hasta el final de la línea (hasta que veas el cuadradito al final de la línea).

Ahora, lo que queremos hacer es que el Amstrad imprima lo que hay en el bolsillo CAJA, justo tras el mensaje, en la misma línea. Para ello, el truco podría ser

```
30 PRINT"CAJA = " ; CAJA
```

Ahora pulsa el signo ; y pulsa la tecla SPACE para dejar un espacio de separación.

Ahora escribe CAJA. Pulsa ENTER para indicarle al Amstrad que has terminado de escribir la línea 30.

¿Qué pasa con la línea 40? Ya no la necesitamos porque su trabajo lo realiza la línea 30. Para eliminarla, escribe simplemente 40 y pulsa ENTER. Si quieres, limpia la pantalla con CLS y escribe LIST.

Ahora lanza el Programa de Dinero y comprueba el resultado.

Unicamente por curiosidad, lista la línea 30 y EDITALA de nuevo, pero utilizando esta vez una coma (,) en vez de ;.

Lanza esta versión del programa y observa cómo aparece el resultado esta vez. Ahora lista la línea 30 y vuélvela a dejar como estaba antes.

¿Podemos hacer lo mismo con las líneas 60 y 70? Por supuesto que sí. El ; o la , le indican al Amstrad que imprima lo que viene a continuación en la misma línea de la pantalla. ¿Por qué no modificas la línea 60 y eliminas la 70?

Tu nuevo Programa de Dinero deberá quedar así:

```
10 PEN 1
20 CAJA = 0
30 PRINT"CAJA = " ; CAJA
45 FOR K = 1 TO 20
50 CAJA = CAJA + 5
55 NEXT K
60 PRINT"CAJA = " ; CAJA
80 END
```

BAILANDO CLAQUE

Me pregunto si habrás visto alguna vieja película musical en la que alguien baila sobre una pista de baile dividida en cuadrados igual que un tablero de ajedrez.

Imagina que la pantalla es como esa pista de baile de tablero de ajedrez.

La "pista de baile" de la pantalla está dividida en 40 casillas yendo desde la izquierda hacia la derecha. Cuando vas de arriba abajo, la "pista de baile" se divide en 25 casillas. Si quieres, puedes dividir un pedazo grande de papel en casillas. Si lo haces, hazlo de 40 casillas de ancho y 25 casillas de alto. Esto te puede ayudar a imaginar mucho mejor lo que estoy diciendo. Cada casilla puede incluir una sola cosa. Puede tratarse de un carácter como A o M, o un número como 5 o 9.



Cada casilla, además, se divide en 8 PIXELS en sentido vertical y 8 PIXELS en sentido horizontal. Los PIXELS son pequeños bloques.

Volvamos al Programa de Dinero e imagina que queremos mover la palabra CAJA a través de las casillas de la pantalla.

La instrucción para mover cosas a través de la pantalla es TAB.

Supongamos que queremos mover CAJA cinco casillas. TAB (5) lo hará.

Puedes imaginar que hay un bailarín llamado CURSOR DE TEXTO que lleva CAJA a través de la pista de baile de la pantalla. El número 5 entre paréntesis le indica al CURSOR DE TEXTO que mueva CAJA cinco casillas.

EDITA la línea 30 del Programa de Dinero:

```
30 PRINT TAB (5);"CAJA = " ; CAJA
```

Antes de seguir, pensemos cómo deberíamos EDITAR la línea 30.

Con la tecla SHIFT y la flecha, sube hasta la línea 30. Después, con la tecla COPY vete hasta que el cuadradito esté después de la T de PRINT.

Ahora pulsa un espacio, y escribe TAB (5); después pulsa otra vez la tecla COPY hasta el final de la línea.

Ahora pulsa ENTER.

Ahora lista y lanza el programa.

Supongamos que queremos que el CURSOR DE TEXTO baile a través de la pantalla, pero no solamente cinco cuadros en dirección horizontal, sino también diez cuadros en dirección vertical.

LOCATE 5,10 lo hará. El primer número es 5, le dice al CURSOR DE TEXTO cuantos cuadros horizontales debe recorrer mientras baila. El segundo número es 10, le dice al CURSOR DE TEXTO cuantos cuadros tiene que bajar mientras baila.

¿Dónde empieza su baile el CURSOR DE TEXTO?

Mira la esquina superior izquierda de la pantalla. Imagina que el CURSOR DE TEXTO esta allí, vestido con sombrero de copa, frac y zapatos brillantes.

Si le dices TAB (5), bailará por la pantalla á través de cinco casillas. Una vez hecho esto, soltará lo que lleve.

Si, en vez de eso, le dices LOCATE 10,5, bailará a través de 10 cuadros horizontales y 5 cuadros hacia abajo. Luego soltará lo que lleve en ese punto de la pantalla.



Ahora modifiquemos la línea 30:

```
30 LOCATE 5,10 : PRINT"CAJA=" ; CAJA
```

Lista y lanza el programa.
Modifiquemos ahora la línea 60:

```
60 LOCATE 5,12 : PRINT"CAJA" ; CAJA
```

¿Dónde crees que bailará el CURSOR DE TEXTO si le decimos LOCATE 5,12?
Lista y ejecuta el programa.

LOS COLORES DEL ARCO IRIS



Hasta ahora has escrito en la pantalla con tinta amarilla sobre papel azul. Al estar la tinta amarilla sobre el papel azul decimos que el color de arriba es el COLOR DE LOS CARACTERES y el color de abajo es el COLOR DE FONDO.

Recuerda, cualquier cosa que esté encima se dice que es relativa a los CARACTERES. Cualquier cosa que esté debajo se dice que es relativa al FONDO.

Si queremos escribir con tinta roja, podemos decírselo al Amstrad escribiendo PEN 3.

Hay 27 colores en el Amstrad, de los cuales se pueden usar a la vez:

En MODE 0, 16 colores de caracteres, 16 colores de fondo.

En MODE 1, 4 colores de caracteres, 4 colores de fondo.

En MODE 2, 2 colores de caracteres, 2 colores de fondo.

LISTA DE COLORES

1 Negro	14 Azul pastel
2 Azul	15 Naranja
3 Azul brillante	16 Rosa
4 Magenta	17 Magenta pastel
5 Malva	18 Verde brillante
6 Rojo brillante	19 Verde mar
7 Púrpura	20 Ciano brillante
8 Magenta brillante	21 Lima
9 Verde	22 Verde pastel
10 Ciano	23 Ciano pastel
11 Azul cielo	24 Amarillo brillante
12 Amarillo	25 Amarillo pastel
13 Blanco	26 Blanco brillante

Supongamos que queremos que CAJA aparezca en tinta ROJA (caracteres) sobre papel AMARILLO (fondo). Escribe una nueva línea, la Línea 10:

```
10 PEN 3
```

De este modo obtenemos tinta ROJA (CARACTERES). Escribe otra nueva línea, la Línea 12:

```
12 PAPER 1
```

De este modo obtenemos papel AMARILLO (FONDO).

Lista el programa y, luego, lánzalo.

¡Espera un minuto! Es verdad que podemos ver las letras rojas sobre fondo amarillo, pero el resto de la pantalla no es amarillo. Aunque lo que vemos es bonito y debemos recordarlo, lo que ahora debemos hacer es limpiar la pantalla completamente para ponerla del color del papel (fondo). El comando CLS lo hace.

Lista el programa y escribe esto:

```
15 CLS
```

Ahora lanza el programa y mira lo que ocurre.

Supongamos que modificamos la línea 12 de modo que ponga PAPER 3, para obtener un fondo de papel ROJO. ¿Qué ocurre si escribimos en tinta roja sobre papel rojo? ¿Por qué no lo intentas y miras lo que ocurre al lanzar el programa?

Si lo pruebas, intenta también listar el programa.

Si quieres modificar el color del fondo directamente, sin escribir una línea de programa, lo único que debes hacer es escribir PAPER y el número del color de fondo deseado.

Así, si ves que no puedes ver el listado porque la tinta y el papel son los dos del mismo color, cambia el color del fondo. Puedes hacerlo directamente utilizando PAPER con un número distinto al número de la tinta de los caracteres. Prueba con PAPER 0. Ahora puedes listar el programa y verlo de nuevo.

Pero, si quieres tener colores de caracteres y de fondo diferentes al ejecutar el programa, debes asegurarte de que las líneas 10 y 12 contengan diferentes colores.

Prueba con distintos colores de fondo y de caracteres.

PARPADEO

El Amstrad te permite que los colores parpadeen. Prueba esto: INK 1, 6, 24 y la tecla RETURN. ¿Qué ocurre? Has llenado el lápiz núm.1 con tinta roja (6) y amarilla (24). Prueba ahora esto, INK 1, 18, 15. Y después prueba esto INK 1,

24, 24. Como las tintas son iguales, el color no parpadea.

El borde puede cambiar de color (de 0 a 27)

Escribe

16 BORDER 15

y lanza otra vez el programa.

DULCES 16

El Amstrad tiene 27 colores de fondo y 27 colores de caracteres.

Los números de lápiz (caracteres) pueden ir desde 0 a 15. Los números de papel (fondo) pueden ir desde 0 a 15.

Los números del borde pueden ir de 0 a 27.

Por supuesto, tendrás que EDITAR el programa si deseas probar con nuevos colores.

Si quieres EDITARLO, primero lístalo. Ahora debería estar así:

```
 5 MODE 1 : BORDER 24
10 PEN 3
12 PAPER 1
15 CLS
20 CAJA = 0
30 LOCATE 5,10 : PRINT"CAJA =" ; CAJA
45 FOR K = 1 TO 20
50 CAJA = CAJA + 5
55 NEXT K
60 LOCATE 5,12 : PRINT"CAJA =" ; CAJA
80 END
```

Ahora pon una cinta en tu grabador de cassettes y salva este programa con otro nombre.

Modifica la línea 10 del modo siguiente:

```
10 INK 1, 18
```

Modifica la línea 12 así:

```
12 PAPER 2 : BORDER 6
```

Ahora lanza el programa.

Puedes pasar un rato distraído utilizando distintos números para el fondo y los caracteres.

DE VUELTA A LOS BOLSILLOS DEL AMSTRAD

Si lo recuerdas, los bolsillos de la memoria del Amstrad pueden contener CADENAS. Para mostrar que lo que hay dentro del bolsillo es una cadena, podemos imaginar que la etiqueta adhesiva del bolsillo lleva escrito el nombre del bolsillo, con un signo $ al final del nombre.

Así que supongamos que tenemos un bolsillo de cadenas que hemos etiquetado A$. Supongamos que queremos poner la cadena "CARA" en su interior.

Ahora vamos a escribir algo nuevo, así que no utilizaremos el Programa de Dinero. Si dicho programa está en el Amstrad, escribe NEW.

Escribe:

```
10 CLS
20 A$ = "CARA"
```

Así colocamos la cadena de caracteres "CARA" en el bolsillo de variables de cadena etiquetado como A$.

Escribe:

```
30 PRINT A$
```

Lanza este pequeño programa.

No olvidemos a nuestro amigo el CURSOR DE TEXTO, el bailarín del TAB.



Lista el programa y modifica la línea 30 de este modo:

```
30 LOCATE 5,10:PRINT A$
```

La línea 30 muestra en la pantalla el contenido del bolsillo de la variable de cadena A$. ¿Qué contiene A$?

Prueba a escribir en tinta roja (caracteres) sobre papel amarillo (fondo).

¿Lo has logrado? He aquí lo que tenemos:

```
10 CLS
20 PEN 3
30 PAPER 1
40 A$ = "CARA"
50 LOCATE 5,10:PRINT A$
60 END
```

¿Por qué no escribes este programa y lo pruebas?

Ahora añadamos:

```
55 LOCATE 5,12:PRINT "COL"
```

Lancemos el programa.

Bueno, el contenido del bolsillo de la variable A$ no ha cambiado. Todavía tenemos "CARA" en él. Pero prueba a hacer esto:

```
57 A$ = A$ + "COL"
58 LOCATE 5,14:PRINT A$
```

Ahora listémoslo y lancémoslo.

¿Qué contiene ahora A$?

Salva el programa del modo habitual. Salvémoslo con el nombre "BICHOS".

¿Qué significa el signo + en la línea A$ = A$ + "COL"? Lo que significa es que "CARA" y "COL" se unen para formar "CARACOL". A

continuación, "CARACOL" es colocado en el bolsillo de la variable de cadena A$. El bolsillo A$, que utilizábamos para contener "CARA". Pero ahora su contenido ha cambiado a "CARACOL".

El poner dos cadenas unidas de modo que se forme una nueva cadena se llama CONCATENACION.

Prueba a escribir directamente, y no en una línea de programa:

```
PRINT"SOL" + "DADO"
```

Ahora escribe NEW y prueba este pequeño programa:

```
10 A$ = "SOL"
20 B$ = "DADO"
30 C$ = A$ + B$
40 LOCATE 10,20:PRINT C$
50 END
```

Si miras las líneas 10, 20 y 30, verás que ahora tenemos tres bolsillos de variables de cadena. Estos bolsillos son los llamados A$, B$ y C$.

La línea 10 pone "SOL" en el bolsillo A$.

La línea 20 pone "DADO" en el bolsillo B$.

La línea 30 busca lo que hay en ambos bolsillos A$ y B$. Copia "SOL" y "DADO" y lo junta para formar "SOLDADO". Pone "SOLDADO" en el bolsillo C$.

Luego, para terminar, la línea 40 nos muestra, en la pantalla, lo que contiene el bolsillo C$.

SI NO LE EXPLICAS LO QUE QUIERES DECIR, EL AMSTRAD NO SABE LO QUE QUIERES DECIR



Solo para ver lo que ocurre, escribe NEW y, luego, escribe directamente:

```
PRINT C$
```

El Amstrad no te comprende porque no le has dicho previamente que cree el bolsillo de la variable de cadena C$. Tienes que escribir antes algo parecido a la Línea 10 del programa anterior. De modo que sería correcto si escribieses, por ejemplo:

```
C$ = "CAJA"
PRINT C$
```

De nuevo, para ver qué pasa, escribe esto:

```
PRINT CAJA
```

El Amstrad no te comprende.

Recuerda, para ser una cadena de caracteres, CAJA debe ir entre comillas ". "CAJA" es una cadena de caracteres.

Para ser un bolsillo de variable de cadena, capaz de contener una cadena, C debe llevar al final el signo $. C$ es una variable de cadena.

Por lo tanto, si escribes únicamente PRINT CAJA, el Amstrad piensa que debe buscar un bolsillo de variable NUMERICA llamado CAJA. Por eso el Amstrad imprime 0 Pero si antes le decimos al Amstrad que hay un bolsillo de variable numérica llamado CAJA, se pondrá bastante contento. Así que escribe:

```
CAJA = 30
PRINT CAJA
```

¿Qué ves en la pantalla?

De paso, diremos que los mensajes que te da el Amstrad cuando no puede comprender algo de lo que has escrito, se llaman MENSAJES DE ERROR.

CONCATENACION DE CADENAS Y SUMA DE NUMEROS

CONCATENACION es una palabra muy, muy larga. Es un bocado muy grande, pero su significado es muy simple. Simplemente significa poner caracteres uno junto a otro. Así, si pones "ARMA" junto a "RIO", tendrás "ARMARIO".





Recuerda que debes decirle a nuestro Amstrad exactamente lo que debe hacer. Para poner las cadenas una junto a otra tienes que colocar el signo + entre ellas. De otro modo, el Amstrad no comprenderá lo que quieres decirle.

Así, para decirle al Amstrad que ponga las cadenas "ARMA" y "RIO" una junto a otra, deberás utilizar algo parecido a D$ = "ARMA" + + "RIO". De este modo, el bolsillo D$ contendrá la cadena de caracteres "ARMARIO".

Por supuesto, si quieres que el Amstrad muestre lo que hay en el bolsillo D$, tienes que decirle PRINT D$.

Ahora, veamos algo interesante. Escribe:

```
PRINT 3 + 4
```

La respuesta que aparece es 7. Aquí el signo + no coloca el 3 junto al 4 para obtener 34. En vez de eso, nos da 7.

¿Por qué ocurre eso? Bueno, eso es porque no hay comillas " alrededor del 3 y del 4, y de ese modo, el Amstrad sabe que se trata de NUMEROS en vez de cadenas. Por lo tanto, SUMA los números 3 y 4 y nos da el resultado de 7.

¡Ajá! ¿Puedes adivinar lo que ocurriría si pusieras las comillas " alrededor del 3 y del 4? ¿Por qué no lo pruebas y lo ves?. Escribe:

```
PRINT"3" + "4"
```

CORTANDO LA CADENA

Espero sinceramente que hayas salvado "BICHOS" en cinta. Si lo hiciste, puedes cargar "BICHOS" del modo habitual.

Este es el listado:

```
10 CLS
20 PEN 3
30 PAPER 1
35 CLS
40 A$ = "CARA"
50 LOCATE 5,10 : PRINT A$
55 LOCATE 5,12 : PRINT "COL"
57 A$ = A$ + "COL"
58 LOCATE 5,14 : PRINT A$
60 END
```

Coge un pedazo de papel y escribe en el: CARACOL.

Ahora, empezando por la izquierda, a partir de la letra C, cuenta uno-dos-tres-cuatro hasta que llegues a la letra A. ¿Qué es lo que tienes? Exacto, CARA.

Si cortas el papel justo tras la A, tendrás dos pedazos de papel. El pedazo de la izquierda pondrá CARA.

(De paso, ten cuidado si utilizas tijeras. Pídele permiso a alguna persona mayor antes de utilizarlas. Las cosas que cortan son peligrosas.)

Podemos decirle al Amstrad que haga lo mismo con las cadenas de caracteres. Podemos decirle que corte una cadena del mismo modo que hemos cortado CARACOL, empezando por la izquierda.

La palabra especial que lo hace es LEFT $.

Puedes imaginar que el signo (es como un par de tijeras dispuestas a cortar la cadena.

Añade estas líneas al programa BICHOS:

```
60 L$ = LEFT $ (A$,4)
65 LOCATE 5,16 : PRINT L$
```



Veamos la línea 60. El Amstrad busca en el bolsillo A$ y encuentra en su interior "CARACOL". Entonces cuenta cuatro caracteres, empe-

zando por el primero de la izquierda, la C. Tras contar los cuatro caracteres, tiene la cadena CARA, y la coloca en el bolsillo de la variable de cadena L$.

Luego la línea 65 nos muestra, en la pantalla, lo que contiene L$.

Unicamente por curiosidad, prueba a contar desde la derecha con la palabra ARMARIO.

Escribe ARMARIO en un pedazo de papel. Empieza por la derecha, desde la letra O. Ves hacia atrás, contando uno-dos-tres. Has llegado a la letra R. ¿Qué es lo que tienes? Muy bien, RIO.

Si cortas el papel justo antes de la R, tendras dos pedazos. El pedazo de la derecha pondra RIO.

Añadamos las líneas 70, 75 y 80 a BICHOS:

```
70 R$ = RIGHT $ (A$,3)
75 LOCATE 5,20 : PRINT R$
80 END
```

Veamos la línea 70. El Amstrad busca dentro del bolsillo A$ y encuentra la cadena de caracteres "CARACOL". Entonces cuenta tres caracteres hacia atrás, empezando por la L. Obtiene la cadena "COL", y la coloca en el bolsillo de la variable de cadena R$.

La línea 75 nos muestra, en la pantalla, el contenido de R$.

El programa ha quedado bastante cambiado, de manera que ¿por qué no lo salvamos de nuevo? Llámalo BICHOS1 si quieres.

QUE DEBO HACER SI...

¿Recuerdas nuestra lista de cosas para hacer en el fin de semana?

```
10 JUGAR AL FUTBOL
20 LEER UN LIBRO
30 VER LA TELEVISION
40 COMER
50 JUGAR CON EL AMSTRAD
60 END
```

Añadamos la línea 15:

```
15 TOMAR UN REFRESCO
```

Ahora, cuando leas la lista de cosas a hacer, tendrás un refresco tras el partido de fútbol. Pero, quizá en alguna ocasión no tengas sed. Por lo tanto, solo querrás tomar un refresco cuando realmente tengas sed.

Modifiquemos la línea 15 de este modo:

```
15 SI TENGO SED, ENTONCES TOMARE
   UN REFRESCO
```

Antes, debías tomar siempre un refresco. Ahora, como has utilizado SI y ENTONCES, tomarás un refresco únicamente cuando tengas sed.

SI y ENTONCES (en inglés IF y THEN) son palabras que el Amstrad puede comprender.

HOMBRE RICO, HOMBRE POBRE

Imagina que empiezas con 2 pesetas en tu bolsillo de CAJA. Supongamos que una persona muy generosa te da 5 pesetas para que las pongas en tu bolsillo de CAJA. Supongamos que lo hace diez veces. Tan pronto como tu bolsillo de CAJA contenga más de 20 pesetas, tú gritas "¡SOY RICO! ¡SOY RICO!".

Escribámoslo con cuidado:

MI BOLSILLO DE CAJA CONTIENE 2 PESETAS.

Otro bolsillo contiene 10, para mostrar las veces que la persona generosa me da dinero para añadir a la CAJA. A este bolsillo lo llamaré KONTADOR.

Seguiremos añadiendo 5 a CAJA durante 10 veces, ya que KONTADOR contiene 10.

Si CAJA llega a ser superior a 20, gritaré "¡SOY RICO! ¡SOY RICO!".

Pero, de cualquier modo, tanto si tengo más de 20 pesetas o no, indicaré cuánto hay en el bolsillo de CAJA.

Por supuesto, tendré que comprobar cuándo he añadido 10 veces el dinero. Ahora probemos a escribir NEW y hacer un programa que el Amstrad pueda comprender:

```
10 PEN 3
20 CAJA = 2
30 FOR KONTADOR = 1 TO 10
40 CAJA = CAJA + 5
50 IF CAJA > 20 THEN PRINT"¡SOY RICO!
   ¡SOY RICO!"
60 PRINT"CAJA = " ; CAJA
70 NEXT KONTADOR
80 END
```

Veamos la línea 50. ¿Adivinas lo que significa el signo > en IF CAJA > 20? Correcto, el signo > significa simplemente mayor que.

(Hay otro signo similar, <, colocado al lado del signo mayor que. Significa menor que).

Lanza el programa.

¿Cuánto dinero contiene el bolsillo CAJA al final?

Bien, CAJA termina conteniendo 52.

¿Cuál es la utilidad de IF? Es muy útil. Te permite hacer algo únicamente cuando hay un motivo. Te permite tomar una decisión. IF tienes sed, THEN decides tomar un refresco. Solamente llevarás a cabo la línea 15 de tu lista de fin de semana si tienes la razón correcta para hacerlo.

De modo similar, a causa del IF de la línea 50, únicamente gritarás que eres rico cuando tengas más de 20 pesetas en el bolsillo.

Para verlo, cambia la línea 40 de este modo:

```
40 CAJA = CAJA + 1
```

Ahora, el bolsillo de CAJA se incrementa en 1 cada vez, en vez de 5.

Lanza el programa y mira lo que ocurre.

PERO SOLO HASTA...

Supongamos ahora que la persona generosa no quiere ser demasiado generosa.

Supongamos que solo quiere darte 5 pesetas hasta que tu bolsillo de CAJA contenga más de 20 pesetas. Entonces quiere detenerse.

Imagínalo tu mismo. Tienes dos pesetas en el bolsillo de CAJA. El te da 5 pesetas. Tú las añades a las 2 de tu bolsillo de CAJA. ¿Tienes más de 20 pesetas? No. Por lo tanto, te da otras 5 pesetas. Y así sucesivamente. Repetirá la misma operación pero únicamente hasta que tengas más de 20 pesetas.



IF y THEN son dos palabras que el Amstrad comprende.

Escribamos un pequeño programa utilizando IF y THEN. Recuerda utilizar NEW para indicarle al Amstrad que vas a escribir un nuevo programa.

Escribe el programa siguiente. Verás huecos en él, pero no intentes copiarlos. Simplemente escribe las líneas como siempre, una tras otra. Por supuesto, cuando termines de escribir una línea, debes utilizar ENTER para que el Amstrad lo sepa.

```
 10 CLS
 20 PEN 1
 30 PAPER 0
 40 PRINT"¡SOY POBRE!"
 50 CAJA = 2
 80 CAJA = CAJA + 5
 90 PRINT"CAJA = " ; CAJA
100 IF CAJA > 20 THEN GOTO 110
105 GOTO 80
110 PRINT
120 PRINT
130 PRINT"ERES RICO."
140 PRINT"POR LO TANTO, ¡NO HAY MAS DI-
    NERO PARA TI!"
150 END
```



Mira las líneas 110 y 120 del programa. Estas líneas sólo imprimen líneas en blanco. Las líneas en blanco separan el mensaje "ERES RICO" de los otros mensajes de la pantalla. Lanza el programa y mira lo que ocurre. Luego, si quieres, elimina las líneas 110 y 120 y mira lo que ocurre cuando lanzas de nuevo el programa.

¿Recuerdas cómo se elimina una línea completa? Simplemente escribe el número de línea y pulsa ENTER.

De paso, estoy seguro de que habrás notado lo siguiente:

FOR siempre va unido a un NEXT.

IF siempre va unido a un THEN.

Van siempre juntos como las dos rebanadas de un bocadillo. Y, al igual que un bocadillo habitualmente lleva algo entre las dos rebanadas, también habitualmente hay algo entre ellos.

EL AMSTRAD ESPERA...



Si estás esperando una carta el día de tu cumpleaños, es posible que te levantes temprano. Quizás te quedes esperando al lado de la puerta, esperando que el cartero introduzca una carta en el buzón.

Del mismo modo, puedes hacer que el Amstrad espere a que tú introduzcas algo en uno de

los bolsillos de su memoria.

¿Qué palabra necesitará el Amstrad para introducir algo en un bolsillo?

La palabra INPUT.



El Amstrad puede esperar a que le des un número para colocarlo en un bolsillo de variable numérica. O puede esperar a que le introduzcas una variable de CADENA de caracteres en un bolsillo de variable de cadena.

¿Cómo sabe qué tipo de variable debe introducir en cada bolsillo?

Bueno, una vez más, todo consiste en ver si hay o no un signo $, que es el que le indica al Amstrad que debe esperar una variable de cadena de caracteres.

Por ejemplo, si le dices al Amstrad que haga INPUT N$, pensará que lo que escribes es una cadena de caracteres. Lo pondrá en el bolsillo de variable de cadena N$.

Pero si le dices al Amstrad que haga INPUT N, pensará que lo que escribes es un número. Pondrá ese número en el bolsillo de variable numérica N.

PIENSA UN NUMERO

Ahora escribe NEW y, a continuación, escribe este corto programa:

```
 10 CLS
 20 PEN 2
 30 LOCATE 3,2 : PRINT "ESCRIBE"
 40 LOCATE 3,3 : PRINT "TU NUMERO"
 50 LOCATE 3,4 : PRINT "FAVORITO"
 60 LOCATE 3,5 : PRINT "DESDE 1 A 9"
 70 INPUT NUM
 80 LOCATE 1,7 : PRINT "TU FAVORITO ES"
 90 LOCATE 8,10 : PRINT NUM
100 NUM = NUM*9
110 RESPUESTA = NUM*12345679
120 LOCATE 3,12 : PRINT "AQUI TIENES UN
    MONTON DE"
130 PEN 1
140 LOCATE 6.14 : PRINT RESPUESTA
150 INK 1,24,6
160 LOCATE 6,4 : PRINT "!!!!"
170 END
```

Ejecuta el programa. Ahora lístalo.

Veamos la línea 70. Si hubieras escrito NUM = 8 en la línea 70, el número 8 ya habría estado en el bolsillo de la variable numérica NUM. Tendrías que limitarte a ese valor y no podrías elegir un número entre 1 y 9.

Pero, como la línea 70 lleva la palabra INPUT, el Amstrad espera a que tú escribas un número, que es el que pondrá en el bolsillo NUM.

Para indicarte que está esperando que le digas algo, el Amstrad muestra un signo ? y un cuadradito en la pantalla.

Veamos la línea 100. Aquí, lo que está contenido en el bolsillo NUM se multiplica por 9.



Supongamos que escribes un 4 como número favorito. El bolsillo NUM contendrá el 4. Tras hacer lo que dice la línea 100, el bolsillo NUM del Amstrad contendrá 36.

Veamos la línea 110. Aquí, NUM se multiplica por un número muy largo. Pero no tenemos que preocuparnos porque el Amstrad hará este trabajo por nosotros. Pondrá el nuevo número en el bolsillo RESPUESTA.

Ya deberías comprender todo lo que hacen las

demás líneas del programa. Pero veamos una vez más el programa línea por línea. Mira el listado en la pantalla a medida que lees el parrafo siguiente.

La Línea 10 le dice al Amstrad que limpie la pantalla.

La Línea 20 pone el número de color 2 para los CARACTERES.

La Línea 30 utiliza LOCATE para imprimir parte de un mensaje. LOCATE baila a través de 3 espacios en la pantalla.

La Línea 40 utiliza LOCATE para otra porción de mensaje. Baila 3 espacios a través de la pantalla.

La Línea 50 utiliza LOCATE para otra porción de mensaje. Baila 3 espacios a través de la pantalla.

La Línea 60 utiliza LOCATE para la última parte del mensaje. Baila 3 espacios a través de la pantalla.

La Línea 70 hace que el Amstrad espere a que tú escribas un número. Ese número lo pone en la variable numérica N.

La Línea 80 utiliza LOCATE para mostrarte otro mensaje en la pantalla.

Utilizamos números diferentes para LOCATE porque queremos poner cosas en sitios distintos de la pantalla.

La Línea 90 utiliza PRINT y LOCATE para mostrarnos, en un lado de la pantalla, lo que contiene NUM en este momento.

La Línea 100 multiplica el contenido de NUM por 9. De ese modo NUM pasa a contener 9 veces lo que tenía antes.

La Línea 110 multiplica NUM por un número muy largo. No tenemos que molestarnos en cal-



cularlo porque el Amstrad lo hace por nosotros. Ese número lo pone en la variable numérica RESPUESTA.

La Línea 120 coloca otro mensaje en la pantalla, utilizando LOCATE.

La Línea 130 cambia el número de color de los caracteres a 3. Todo lo que se imprima a continuación saldrá en este nuevo color.

La Línea 140 utiliza PRINT y LOCATE para mostrar el contenido de RESPUESTA, en un lado de la pantalla.

La Línea 150 modifica el número del color de los CARACTERES a 1. Al escribir INK 1,24,6, la tinta parpadeará.

La Línea 160 imprime únicamente cuatro signos !!!! en un lado de la pantalla, utilizando LOCATE.

¡QUE NOTABLE!

Hemos visto todas las líneas del programa pero, por supuesto, tendremos que saber mejor las cosas y necesitaremos menos explicaciones.

Si queremos poner algunas pequeñas explicaciones en el propio programa, de manera que lo pueda comprender cualquier otro, podemos utilizar REM. REM simplemente le dice al Amstrad que lo que va a continuación, en la línea, es una nota o una observación. Una observación te indica algo sobre una parte del programa. El Amstrad no hace nada cuando, durante la ejecución de un programa, se encuentra un REM. La línea REM únicamente aparece cuando listas el programa.

Por ejemplo, podemos añadir la Línea 145, antes de la Línea 150:

```
145 REM INK 1,24,6
```

Podemos añadir otras líneas REM del mismo modo, si queremos explicar alguna parte importante de un programa. Es muy útil en un programa realmente largo, que podría ser difícil de leer y de comprender sin una explicación de los puntos más importantes.

Pero, recuerda, demasiadas líneas REM tambien pueden hacer que un programa sea difícil de leer. Tienes que utilizarlas con sentido común.

INSPIRA Y SOPLA

Tengo que admitirlo, ha pasado mucho tiempo desde que iba a la escuela. Pero me parece recordar una historia referente a tres cerditos que construyeron una casa cada uno. No puedo recordarla completamente, pero estoy seguro de que uno de los cerditos construyó su casa con paja y otro la construyó con ladrillos. Y había un gran lobo malvado que inspiró y sopló y derribó la casa de paja. Pero quedó completamente agotado de tanto inspirar y soplar cuando intentó derribar la casa de ladrillos.



¿Qué tiene que ver esto con los programas? Tiene mucha relación.

Una casa de paja es sucia. Se cae a pedazos y tienes que estar siempre remendándola. Si hay un pedazo de pared que necesita una reparación, tienes que quitar tambien pedazos de pared en buen estado y la paja que pongas es muy difícil que encaje exactamente en el lugar donde quieras ponerla.

Si tu casa está construida de ladrillos, es fácil quitar un ladrillo estropeado y poner uno nuevo. Es fácil reemplazar una parte por otra.

Del mismo modo, debes escribir tus programas de modo que sean limpios y construirlos a base de pedazos pequeños. Cada pedazo debe tener un propósito claro. Un ladrillo tiene una forma sólida y clara. encaja limpiamente con otros ladrillos.

¿Es más fácil decirlo que hacerlo? Al principio, sí. Pero, con la práctica, llegará a ser tan fácil hacerlo como decirlo.

Quiero escribir un juego.

1. Primero, quiero PREPARAR LA PANTALLA.
2. Luego quiero decirle al Amstrad que cree un bolsillo de variable de cadena y un bolsillo de variable numérica.
3. Quiero darle algunos mensajes al jugador para EXPLICARLE COMO FUNCIONA EL JUEGO.
4. Quiero preguntarle al jugador si quiere jugar o no. Tan pronto como quiera dejar de jugar, el juego terminará. De otro modo continuará el juego.
5. Luego quiero presentarle un problema al jugador. Quiero que adivine si el lobo está "inspirando" o "soplando". De eso es de lo que se trata el juego.
6. Mientras el jugador intenta adivinar lo que está haciendo el lobo, quiero tambien comprobar si lo ha adivinado. ¿Ha GANADO? Si ha ganado, tengo que hacer esto:

Tengo que decirle que ha ganado y, a continuación, terminar el juego.

7. Quiero decirle cuántos intentos lleva hechos. Tambien quiero comprobar si ha hecho mas de 3 intentos. Si los ha hecho, el lobo ha inspirado y soplado suficientes veces como para derribar su casa. Por lo tanto, ha PERDIDO. Tengo que darle un mensaje y terminar el juego.

Mientras el jugador no haya hecho más de 3 intentos, puede seguir jugando hasta que quiera pararse.

Si piensas escribir cualquier tipo de programa, verás que utiliza el mismo tipo de ladrillos de programa:

Un ladrillo para mostrar las cosas en la pantalla al comienzo del programa.

Un ladrillo para crear ciertas variables.

Un ladrillo para dar mensajes referentes al programa.

Un ladrillo para introducir la información que el programa utilice.

Un ladrillo para crear alguna salida de información facilitada por el programa. Puede tratarse de un problema para el usuario, como en un juego. Pero en vez de eso puede ser una solución, como cuando los ordenadores suman una serie de números y nos dan el resultado en la pantalla.

Un ladrillo para ver si el programa tiene un resultado igual al del jugador. En ese caso, ¿ha ganado el jugador?

Un ladrillo para ver si el programa tiene un resultado distinto al del jugador. En ese caso, ¿ha perdido el jugador porque ha utilizado todos sus intentos?

Un ladrillo para mantener informado al usuario de lo que está ocurriendo. En este caso, ¿cuántos intentos.

Cada ladrillo de PROGRAMACION se llama un PROCEDIMIENTO. Observa que no hemos hecho ningún programa. Pero hemos hecho algo más importante que escribir líneas de programa. Hemos planificado un programa para construirlo con ladrillos de PROCEDIMIENTOS.

EL JUEGO DE INSPIRA Y SOPLA

1. Describiremos o DEFINIREMOS el ladrillo del PROCEDIMIENTO DE PREPARACION DE LA PANTALLA. Démosle un nombre para recordarlo con él. Llamémosle PROCPREPARACION. Por lo tanto nuestra descripción se puede llamar REM PREPARACION. Hela aquí:

```
REM PREPARACION

   El color de los caracteres es rojo.
   El color del fondo es amarillo.

RETURN
```

Observa el RETURN. ¿Qué crees que significa? Correcto, significa que hemos terminado la definición de un procedimiento.

2. DEFINAMOS el PROCEDIMIENTO que creá las variables que necesitamos para Inspira y Sopla. Llamemosle VARIABLES.

```
REM VARIABLES
```

Pregunto el nombre del jugador. Es decir, hago un INPUT para poner el nombre en una cadena de caracteres. La pongo en un bolsillo de variable de cadena llamado NOMBRE$.

Le digo al Amstrad que cree un bolsillo de variable numérica llamado JUGADAS conteniendo un 0. (JUGADAS se incrementara en 1 cada vez que el jugador haga una jugada.) JUGADAS le dirá al jugador cuántas jugadas lleva hechas hasta el momento.

RETURN

3. DEFINO el PROCEDIMIENTO que le dice al jugador las REGLAS del juego.

REM REGLAS

Imprimo "Hola" (y el nombre).
Imprimo "El juego de Inspira y Sopla es bonito y difícil."
Imprimo "¿Inspiro o Soplo?"
Imprimo "escribe INSPIRAS o SOPLAS y pulsa ENTER.
Imprimo "Suyo afectísimo, el Lobo."

RETURN

Ahora tenemos que continuar repitiendo el problema (lo cual forma el ladrillo de procedimiento número 5) HASTA que el jugador escriba "P" para parar el juego. El número 8 es el que lo hace.

Así que "escuchamos" simplemente diciendo IF INKEY$ = "P" THEN END. Vamos comprobando IF INKEY$. Si es igual a "S", el juego se repetirá al llegar a la línea IF INKEY$. También le decimos al jugador que pulse P para parar el juego si es lo que desea. Si quiere seguir jugando, le diremos al jugador que pulse S. Tan pronto como pulse la P, se termina el juego.

4. Este es el procedimiento PROBLEMA. Lo llamaremos así. Tan pronto como el jugador intente resolverlo, incrementaremos JUGADAS en 1.

El problema es: ¿está el Lobo inspirando o soplando? ¿Cómo lo haremos para que el Amstrad elija entre hacer que el Lobo inspire o sople?

Hay una palabra muy útil que se llama RND. Si tenemos los números 1, 2, 3, y así sucesivamente, escritos en pedazos de papel separados, y cerramos los ojos y elegimos uno, estamos haciendo una elección AL AZAR. No puedes estar seguro de cuál es el número que eliges. RND le dice al Amstrad que cierre los ojos y elija un número.

Si pones 1 + INT(RND*12), elegirá un número entre 1 y 12.

Si pones 1 + INT(RND*2), elegirá 1 o 2. Ya lo tenemos. Vamos a utilizar esto. Si el número que sale es el 1, pondremos "INSPIRAS" en el bolsillo de variable de cadena R$.

Si la función RND elige el 2, pondremos "SOPLAS" en R$.

REM PROBLEMA

Primero, añado 1 a la variable numérica JUGADAS. Así, JUGADAS = JUGADAS + 1.

Hago una variable numérica igual a 1 + INT(RND*2). A esta variable numérica la llamo R. Ahora cambio a rojo el color de los caracteres. Si R es 1, imprimo TAB (1) y LOBO. Si R es 2, imprimo TAB (14) y LOBO.

Ahora, si R = 1, digo R$ = "INSPIRAS". Si R = 2, R$ = "SOPLAS".

Ahora le preguntamos al jugador: "¿estoy inspirando o soplando?".

¿Qué es lo que ha contestado el jugador? Le decimos al Amstrad que espere a recibir INSPIRAS o SOPLAS. Pondremos algo parecido a INPUT A$. Si es "INSPIRAS", lo comprobaremos con R$ para ver si son iguales. Si lo son, le diremos al jugador que ha ganado, utilizando el procedimiento GANADOR. Si R$ no es igual a A$, pasaremos a decirle al jugador cuántas jugadas lleva. Haremos lo mismo en el caso de que el jugador haya escrito "SOPLAS". Si el jugador escribe algo totalmente diferente, simplemente pierde una tirada.

RETURN

5. El procedimiento GANADOR imprime el nombre del jugador, le dice que ha ganado y da un mensaje parecido a "No puedo derribar tu casa. Suyo frustradísimo, Lobo." Luego le decimos al ordenador que termine.

6. Le decimos al jugador cuántas jugadas lleva. Si son más de 3, hacemos un procedimiento que podemos llamar procedimiento INSPIRA-SOPLA. En caso contrario, repetimos el problema.

7. INSPIRA-SOPLA puede imprimir mensajes como "es tu intento número cuatro, tu casa está hecha de paja. He inspirado y soplado y he derribado tu casa. Suyo contentísimo, Lobo". Luego terminamos el juego.

Si el jugador no ha dicho que quería terminar, o no ha ganado, o no ha utilizado todas sus oportunidades, esto significa que INKEY$ no es "P". Por lo tanto, el juego se repite.

Hemos dejado que seas tú mismo el que construyas las partes 5, 6 y 7 como procedimientos.

¡EMPIEZA CON TUS PROGRAMAS!

Hagamos una lista de los procedimientos, junto con otros pedazos del juego Inspira y Sopla:

```
 10 CLS
 20 GOSUB 140 : REM PREPARACION
 30 GOSUB 190 : REM VARIABLES
 40 GOSUB 240 : REM REGLAS
 50 REM
100 GOSUB 340 : REM PROBLEMA
110 GOSUB 550 : REM SIGUE
120 IF INKEY$ = "P"THEN GOTO 850
130 GOTO 100
```

Este es el conjunto del juego de Inspira y Sopla, escrito como una lista de procedimientos. Cada vez que el Amstrad llega a un NOMBRE DE PROCEDIMIENTO, BUSCA DICHO PROCEDIMIENTO, tras la línea 130.

Así, cuando el Amstrad llega a la LINEA 20 y ve PREPARACION busca esto:

```
140 REM PREPARACION
150 PEN 3
160 PAPER 1
170 CLS
180 RETURN
```

Así que sigamos con las definiciones de todos los procedimientos.

```
190 REM VARIABLES
200 PRINT"¿CUAL ES TU NOMBRE?"
210 INPUT NOMBRE$
220 JUGADAS = 0
230 RETURN
240 REM REGLAS
250 PRINT"ESTE ES EL JUEGO DE INSPIRA Y
    SOPLA"
260 PRINT"ESCRIBE INSPIRAS SI CREES
    QUE ESTOY INSPIRANDO"
270 PRINT"ESCRIBE SOPLAS SI CREES QUE
    ESTOY SOPLANDO"
280 PRINT"LUEGO PULSA ENTER"
290 PRINT"SUYO AFECTISIMO, LOBO"
330 RETURN
340 REM PROBLEMA
350 JUGADAS = JUGADAS + 1
360 R = 1 + INT(RND*2)
370 IF R = 1 THEN R$ = "INSPIRAS"
380 IF R = 2 THEN R$ = "SOPLAS"
390 PRINT"¿ESTOY INSPIRANDO?"
400 PRINT"¿ESTOY SOPLANDO?"
410 INPUT B$
420 IF B$ = R$ THEN GOSUB 450 : REM GA-
    NADOR
440 RETURN
450 REM GANADOR
460 PRINT NOMBRE$
470 PRINT"HAS GANADO"
480 PRINT"NO PUEDO DERRIBAR TU CASA"
490 PRINT"SUYO FRUSTRADISIMO"
500 PEN 2
510 PRINT"LOBO"
```

```
520 GOTO 850
540 RETURN
550 REM SIGUE
560 PRINT"TU NUMERO DE JUGADAS HASTA
    AHORA ES"
570 PRINT JUGADAS
580 IF JUGADAS > 3 THEN 600 : REM
    INSPIRA-SOPLA
590 RETURN
600 REM INSPIRA-SOPLA
610 PEN 3
620 CLS
630 PRINT"TU JUGADA NUMERO CUATRO"
640 PRINT"SIGNIFICA QUE TU CASA ES DE
    PAJA"
650 PRINT"HE INSPIRADO, HE SOPLADO Y
    HE DERRIBADO TU CASA"
660 PRINT"SUYO CONTENTISIMO"
670 PEN 2
680 PRINT"LOBO"
700 GOTO 850
710 RETURN
850 REM
860 PRINT"ESCRIBE P PARA PARAR"
870 PRINT"ESCRIBE S PARA SEGUIR"
880 IF INKEY$ = "S"THEN GOTO 10
890 IF INKEY$ = "P"THEN END
895 GOTO 880
```

Escribe el programa. Si quieres listarlo, escribe LIST y pulsa la tecla ENTER. Si quieres parar el listado pulsa ESC una vez, si quieres ver el resto del programa pulsa cualquier otra tecla.

Si ejecutas el programa, observarás que la pantalla no queda muy clara. Es importante que la gente pueda leer fácilmente lo que hay en la pantalla. De manera que trabaja en el programa para hacer que quede un poco mejor. Utiliza todo lo que has aprendido.

Si te encuentras bloqueado en algo, pídele ayuda a alguien que sepa algo de programación. Hay muchas cosas que yo no comprendo y, a menudo, le pido ayuda a otras personas. Pero antes intento hacer las cosas por mí mismo. Por lo tanto, antes que nada, intenta mejorar el programa por tí mismo.



Jugar con un ordenador puede ser tan divertido como instructivo. La idea de este libro es ayudar a los más jóvenes y a sus padres para que comprendan cómo trabajar el Amstrad y como aprovechar al máximo sus posibilidades.
El libro está concebido de una manera sencilla y con un estilo simpático que demuestra que aprender puede ser divertido.
Una gran ventaja de la obra son los capítulos cortos, que invitan a niños y padres avanzar escalonadamente.
Una hora al día con el libro y el ordenador despertarán en el principiante una afición y unas ideas de programación que darán su fruto en el futuro.
Esta obra no es un libro de texto para aprender BASIC pero sí qué es un sistema para conocer los comandos del BASIC del Amstrad.
Niños a partir de siete años pueden entender este libro y por tanto con o sin sus padres podrán hacer uso del ordenador.

EDITORIAL NORAY, S.A.
San Gervasio de Cassolas, 79
08022 Barcelona

AMSTRAD CPC

MÉMOIRE ÉCRITE
MEMORY ENGRAVED
MEMORIA ESCRITA

https://acpc.me/